/*The SAS program (HEI-2010 at the population/group level using NHANES 2009-2010 data (and FPED))

Population Ratio Method.SAS*/

/*INSTRUCTIONS – complete tasks 1-4 in this section, and run these SAS codes before proceeding to the HEI-2010 scoring program that follows*/

/*1. Create a folder on your computer “home folder”, and save the FPED data, NHANES data, Demographic data, and the required HEI-2010 macros in it. Specify the path to the folder. */

%let home = C:\Users\Documents\FPED_NHANES; /*In this Example, the “home” folder is in C Drive, within Documents, and is called FPED_NHANES. */

/*2. Libnames here specify the input files. */

libname NH “&home\NH”; 
libname FPED “&home\FPED”; /*In this Example, the FPED data are in a folder called “FPED”, and the NHANES and Demographic data are in a folder called “NH”, all saved within the “home” folder. These are SAS datasets. */
/*3. Create a folder in the "Home" folder, where the output file, containing HEI-2010 component and total scores are to be exported. Specify the name of the folder. */

filename RES “&home\RES”; /*In this Example, the folder is called “RES”, within the “home” folder, and the exported results will be a csv file called “result”. */
/*4. Read in required HEI-2010 scoring macros. These macros must be saved within the home folder. */

%include “&home\hei2010.beanspeas.allocation.macro.sas”;

%include “&home\hei2010.score.macro.sas”;

/*NOTE: Once you have completed all the steps above, all you need to do is run the SAS program below. Unless you used different names for the folders and datasets, no other action is required from you. */

title 'HEI-2010 scores for NHANES 2009-2010 day 1, AGE >= 2, RELIABLE DIETS, Include Pregnant and Lactating Women';

/*Section (I): Calculations at the individual participant level to obtain variables needed to calculate HEI-2010 scores.*/
/*Step 1: locate the required datasets and variables */

*part a: get FPED data per day;
data FPED;

 set FPED.fped_dr1tot_0910;

run;
*part b: get individual nutrient food intake if reliable recall status;
data NUTRIENT (keep=SEQN WTDRD1 DR1TKCAL DR1TSFAT DR1TALCO DR1TSODI DR1DRSTZ DR1TMFAT DR1TPFAT);

  set NH.DR1TOT_F;

  if DR1DRSTZ=1; /*reliable dietary recall status*/
run;

*part c: get demographic data for persons aged two and older;
data DEMO (keep=SEQN RIDAGEYR RIAGENDR SDDSRVYR SDMVPSU SDMVSTRA);

  set NH.DEMO_F;

  if RIDAGEYR >= 2;

run;

/*Step 2: Combine the required datasets*/

proc sort data=FPED;

  by SEQN;

run;

proc sort data=NUTRIENT;

  by SEQN;

run;

proc sort data=DEMO;

  by SEQN;

run;

data COHORT;

  merge NUTRIENT (in=N) DEMO (in=D) FPED;

  by SEQN;

  if N and D;

run;

/*Step 3: Creates additional required variables: WHOLEFRT, MONOPOLY, ALLMEAT, SEAPLANT, ADDSUGC, SOLFATC, EXALCCAL, and EMPTYCAL10. */
data COHORT;

  set COHORT;

  by SEQN;

  WHOLEFRT=DR1T_F_CITMLB+DR1T_F_OTHER;
  MONOPOLY=DR1TMFAT+DR1TPFAT;

  ALLMEAT=DR1T_PF_MPS_TOTAL+DR1T_PF_EGGS+DR1T_PF_NUTSDS+DR1T_PF_SOY; 

  SEAPLANT=DR1T_PF_SEAFD_HI+DR1T_PF_SEAFD_LOW+DR1T_PF_NUTSDS+DR1T_PF_SOY;

 /**Calculate intake of Empty Calories**/
       ADDSUGC=16*DR1T_ADD_SUGARS;

/*calories from added sugars*/

       SOLFATC=DR1T_SOLID_FATS*9;
/*calories from solid fat*/

      MAXALCGR=13*(DR1TKCAL/1000);  /*max grams of alcohol based on kcal intake*/
      if DR1TALCO <= MAXALCGR then EXALCCAL=0;  /*consumed less than max*/
        else if DR1TALCO > MAXALCGR then EXALCCAL=7*( DR1TALCO - MAXALCGR);  /*get cal from extra alc grams*/
      EMPTYCAL10=ADDSUGC+SOLFATC+EXALCCAL; /*total empty calories in hei2010 definition*/
run; 

/*Section (II): Calculation of weighted means and a variance-covariance matrix and generation of a Monte Carlo dataset, enabling standard errors to be calculated.*/
/*Step 1.  Calculate the weighted means and the variance/covariance matrix for the dietary variables of interest. */
data ONE;

    set COHORT;

    array comp (15) DR1TKCAL DR1T_V_TOTAL DR1T_V_DRKGR DR1T_V_LEGUMES DR1T_F_TOTAL WHOLEFRT DR1T_G_WHOLE DR1T_D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI DR1T_G_REFINED EMPTYCAL10;

    * Turn each variable into an observation of the single variable VBL ;
    * keeping track of the order with the dum_num variable; 

    do i = 1 to 15;

        VBL = comp(i);

        dum_num = i;

        output;

        end;

run;

data ONE;

    set ONE;

    * Create dummies with the same names as the original variables;
    * The i-th dummy gets a 1 if the observation is associated with the

    * i-th original variable;
    array comp (15) DR1TKCAL DR1T_V_TOTAL DR1T_V_DRKGR DR1T_V_LEGUMES DR1T_F_TOTAL WHOLEFRT DR1T_G_WHOLE DR1T_D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI DR1T_G_REFINED EMPTYCAL10;

    do i = 1 to 15;

        if dum_num = i then comp(i) = 1;

        else comp(i) = 0;

        end;

    drop i dum_num;

run;

* Now run PROC (SURVEY)REG with dependent variable VBL and all the dummies;
* as predictors. Force the regression to not include an intercept, and ask;
* for the covariance matrix of the parameter estimates. ;
proc surveyreg data=ONE;

    strata SDMVSTRA;

    cluster SDMVPSU;

    weight WTDRD1;

    model VBL= DR1TKCAL DR1T_V_TOTAL DR1T_V_DRKGR DR1T_V_LEGUMES DR1T_F_TOTAL WHOLEFRT DR1T_G_WHOLE DR1T_D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI DR1T_G_REFINED EMPTYCAL10/noint covb;

    * Output the covariance matrix we wanted all along;
    ods output COVB=CSD_COV;

    title2 "Tricking SURVEYREG into giving us the covariance matrix of means";

run;

* this proc print is for verification and is not necessary;

* uncomment next three lines if more information is desired;

*proc print data= CSD_COV;

*    title2 "Printout of csd_cov dataset -uses complex survey info";

*run;

*If curious about results then remove the noprint option and more ;

* output will be displayed;

proc means data= COHORT n min max mean noprint;

  weight WTDRD1;

  var DR1TKCAL DR1T_V_TOTAL DR1T_V_DRKGR DR1T_V_LEGUMES DR1T_F_TOTAL WHOLEFRT DR1T_G_WHOLE DR1T_D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI DR1T_G_REFINED EMPTYCAL10;

  title2 'look at weighted means';

  output out=WTDM mean= ;

run;

data COVDATA (drop=Parameter);

  set CSD_COV;

  _TYPE_='COV ';

  _NAME_=Parameter;

run;

data WTDM (drop=_TYPE_ _FREQ_);

  set WTDM;

run;

data WTDM;

  set WTDM;

  _TYPE_='MEAN';

run;

data COVDATA;

  set COVDATA WTDM;

run;

* this proc print is for verification and is not necessary;

* uncomment next three lines if more information is desired;

*proc print data=COVDATA;

*  title2 'input to simnorml';

*run;

/*Step 2.  In this step, a Monte Carlo data set with 10,000 rows is generated using the means

 and variance/covariance matrix from step 1*/

*seed value may be changed using a random number generator;

proc simnormal data=COVDATA(type=cov) numreal=10000 seed=51230077 outseed out=SIM_DATA;

  var DR1TKCAL DR1T_V_TOTAL DR1T_V_DRKGR DR1T_V_LEGUMES DR1T_F_TOTAL WHOLEFRT DR1T_G_WHOLE DR1T_D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI DR1T_G_REFINED EMPTYCAL10;

run;

* this proc means is for verification and is not necessary;

* uncomment next four lines if more information is desired;

*proc means data=SIM_DATA n nmiss min max mean stddev;

*    var DR1TKCAL DR1T_V_TOTAL DR1T_V_DRKGR DR1T_V_LEGUMES DR1T_F_TOTAL WHOLEFRT DR1T_G_WHOLE DR1T_D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI DR1T_G_REFINED EMPTYCAL10;

*  title2 "Distributions of Simulated Data";

*run;

* this proc print is for verification and is not necessary;

* uncomment next three lines if more information is desired;

*proc print data= SIM_DATA (obs=20);

*  title2 "Listing of 20 Records from Simulated Data";

*run;

/*Section (III): Allocation of beans and peas (legumes) and application of the HEI-2010 scoring algorithm.*/
/*Step 1: Run the macro to properly allocate the intakes of the FPED variable Legumes in cup equivalents (DR1T_V_LEGUMES) to either Total Protein Foods and Seafood and Plant Proteins (ALLMEAT and SEAPLANT) or Total Vegetables and Greens and Beans (DR1T_V_TOTAL and DR1T_V_DRKGR).  The four resulting variables from this step, named legume_added_DR1T_V_TOTAL, legume_added_BEANGRN, legume_added_ALLMEAT, and legume_added_SEAPLANT are all used in the HEI scoring macro in Step 2*/
%LEG2010A (indat= SIM_DATA,

           kcal= DR1TKCAL,

           allmeat=ALLMEAT,


   seaplant=SEAPLANT,


   v_total=DR1T_V_TOTAL,


   v_drkgr= DR1T_V_DRKGR,


   legumes= DR1T_V_LEGUMES,


   outdat= LSIM_DATA);

/*Step 2: Apply the HEI-2010 scoring macro. */
%HEI2010 (indat= LSIM_DATA,

          kcal= DR1TKCAL,


  lv_total= legume_added_DR1T_V_TOTAL,


  lbeangrn= legume_added_BEANGRN,


  f_total= DR1T_F_TOTAL,


  wholefrt=WHOLEFRT,


  g_whl= DR1T_G_WHOLE,


  d_total= DR1T_D_TOTAL,

       lallmeat= legume_added_ALLMEAT,


  lseaplant= legume_added_SEAPLANT,


  monopoly=MONOPOLY,


  sfat=DR1TSFAT,


  sodi=DR1TSODI,


  g_nwhl= DR1T_G_REFINED,


  Emptycal10=EMPTYCAL10,


  outdat= AFTERMAC);

* this proc means is for verification and is not necessary;

* uncomment next four lines if more information is desired;

*proc means data=AFTERMAC n nmiss min max mean stddev;

*  var legume_added_v_total legume_added_BEANGRN legume_added_allmeat legume_added_SEAPLANT 

    vegden grbnden frtden whfrden wgrnden dairyden

     meatden seaplden faratio sodden rgden sofa_perc;

*  title2 'after legume allocation and hei 2010 scoring macro';

*run;

/*Section (IV): Calculation of HEI-2010 component and total scores and their confidence intervals.*/
/*Step 1.  This step uses univariate and means procedures to compute total and component scores and their standard errors.  */
proc univariate data=AFTERMAC noprint;

  var HEIX1_TOTALVEG HEIX2_GREEN_AND_BEAN HEIX3_TOTALFRUIT HEIX4_WHOLEFRUIT HEIX5_WHOLEGRAIN HEIX6_TOTALDAIRY 

    HEIX7_TOTPROT HEIX8_SEAPLANT_PROT HEIX9_FATTYACID HEIX10_SODIUM HEIX11_REFINEDGRAIN HEIX12_SOFAAS HEI2010_TOTAL_SCORE;

  output out=ci pctlpts=2.5 97.5 pctlpre=h1_ h2_ h3_ h4_ h5_ h6_ h7_ h8_ h9_ h10_ h11_ h12_ totscore_;

run;

proc means data=AFTERMAC noprint;

  var HEIX1_TOTALVEG HEIX2_GREEN_AND_BEAN HEIX3_TOTALFRUIT HEIX4_WHOLEFRUIT HEIX5_WHOLEGRAIN HEIX6_TOTALDAIRY 

    HEIX7_TOTPROT HEIX8_SEAPLANT_PROT HEIX9_FATTYACID HEIX10_SODIUM HEIX11_REFINEDGRAIN HEIX12_SOFAAS HEI2010_TOTAL_SCORE;

  output out=stat min=h1_min h2_min h3_min h4_min h5_min h6_min h7_min h8_min h9_min h10_min h11_min h12_min totscore_min

   max=h1_max h2_max h3_max h4_max h5_max h6_max h7_max h8_max h9_max h10_max h11_max h12_max totscore_max

    mean=h1_mean h2_mean h3_mean h4_mean h5_mean h6_mean h7_mean h8_mean h9_mean h10_mean h11_mean h12_mean totscore_mean

     stddev= h1_stddev h2_stddev h3_stddev h4_stddev h5_stddev h6_stddev h7_stddev h8_stddev h9_stddev h10_stddev h11_stddev h12_stddev totscore_stddev;  

run;

/*Step 2.  This step prepares the results for display*/
data ALL;

  merge CI STAT;

run;

data RESULT (keep=score slabel min max mean stderr lowerci upperci);

  set ALL;

  score='HEIx1          ';

  slabel='HEI-2010 COMPONENT 1 TOTAL VEGETABLES';

  min=h1_min;

  max=h1_max;

  mean=h1_mean;

  stderr=h1_stddev;

  lowerci=h1_2_5;

  upperci=h1_97_5;

  output RESULT;

  score='HEIx2';

  slabel='HEI-2010 COMPONENT 2 GREENS AND BEANS';

  min=h2_min;

  max=h2_max;

  mean=h2_mean;

  stderr=h2_stddev;

  lowerci=h2_2_5;

  upperci=h2_97_5;

  output RESULT;

  score='HEIx3';

  slabel='HEI-2010 COMPONENT 3 TOTAL FRUIT';

  min=h3_min;

  max=h3_max;

  mean=h3_mean;

  stderr=h3_stddev;

  lowerci=h3_2_5;

  upperci=h3_97_5;

  output RESULT;

  score='HEIx4';

  slabel='HEI-2010 COMPONENT 4 WHOLE FRUIT';

  min=h4_min;

  max=h4_max;

  mean=h4_mean;

  stderr=h4_stddev;

  lowerci=h4_2_5;

  upperci=h4_97_5;

  output RESULT;

  score='HEIx5';

  slabel='HEI-2010 COMPONENT 5 WHOLE GRAINS';

  min=h5_min;

  max=h5_max;

  mean=h5_mean;

  stderr=h5_stddev;

  lowerci=h5_2_5;

  upperci=h5_97_5;

  output RESULT;

  score='HEIx6';

  slabel='HEI-2010 COMPONENT 6 DAIRY';

  min=h6_min;

  max=h6_max;

  mean=h6_mean;

  stderr=h6_stddev;

  lowerci=h6_2_5;

  upperci=h6_97_5;

  output RESULT;

  score='HEIx7';

  slabel='HEI-2010 COMPONENT 7 TOTAL PROTEIN FOODS';

  min=h7_min;

  max=h7_max;

  mean=h7_mean;

  stderr=h7_stddev;

  lowerci=h7_2_5;

  upperci=h7_97_5;

  output RESULT;

  score='HEIx8';

  slabel='HEI-2010 COMPONENT 8 SEAFOOD AND PLANT PROTEIN';

  min=h8_min;

  max=h8_max;

  mean=h8_mean;

  stderr=h8_stddev;

  lowerci=h8_2_5;

  upperci=h8_97_5;

  output RESULT;

  score='HEIx9';

  slabel='HEI-2010 COMPONENT 9 FATTY ACID RATIO';

  min=h9_min;

  max=h9_max;

  mean=h9_mean;

  stderr=h9_stddev;

  lowerci=h9_2_5;

  upperci=h9_97_5;

  output RESULT;

  score='HEIx10';

  slabel='HEI-2010 COMPONENT 10 SODIUM';

  min=h10_min;

  max=h10_max;

  mean=h10_mean;

  stderr=h10_stddev;

  lowerci=h10_2_5;

  upperci=h10_97_5;

  output RESULT;

  score='HEIx11';

  slabel='HEI-2010 COMPONENT 11 REFINED GRAINS';

  min=h11_min;

  max=h11_max;

  mean=h11_mean;

  stderr=h11_stddev;

  lowerci=h11_2_5;

  upperci=h11_97_5;

  output RESULT;

  score='HEIx12';

  slabel='HEI-2010 COMPONENT 12 SOFAAS CALORIES';

  min=h12_min;

  max=h12_max;

  mean=h12_mean;

  stderr=h12_stddev;

  lowerci=h12_2_5;

  upperci=h12_97_5;

  output RESULT;

  score='TOTAL HEI 2010';

  slabel='TOTAL HEI-2010 SCORE';

  min=totscore_min;

  max=totscore_max;

  mean=totscore_mean;

  stderr=totscore_stddev;

  lowerci=totscore_2_5;

  upperci=totscore_97_5;

  output RESULT;

run;

/*Step 3 - This step displays the results*/
proc print data=RESULT;

  id score;

  var slabel min max mean stderr lowerci upperci;

  title2 'complex survey design population method - mean and confidence interval of HEI-2010 using NH 09/10 data';

run;

/*Step 4 - This step provides an option to export the results into a CSV file that can be opened in Excel.*/
proc export data=RESULT

  file=RES

  dbms=csv

  replace;

run;

